

HOME Physik Mechanik Elektrizität Optik Atomphysik **Quantenphysik** Publikationen

English


Interferenz - quantenoptisch

Auszug aus einem Maple-Worksheet, in Anlehnung an L. Mandel, Quantum effects in one-photon and two-photon interference, Reviews of Modern Physics, Vol. 71, No. 2, Centenary 1999.


Anmerkungen zu Maple 2018: Das Physics-Package (Maplesoft) erlaubt das Rechnen mit Kets und Bras, ist aber noch verbesserungsfähig, insbes. was das Rechnen in Produkträumen angeht. Der Export von Worksheets nach HTML hat bei Maplesoft nicht die oberste Priorität (in der Nachbearbeitung des Layouts dieses Artikels steckt mehr Arbeit als "in der Physik" :-).

In der Quantenoptik beschreibt man Interferenz durch die Überlagerung von Moden, in denen sich Photonen befinden. Mit einer Mode ist dabei die elektrische Feldstärke als Funktion von Ort und Zeit gemeint (meistens eine ebene Welle) und das Photon wird (mathematisch) durch Operatoren erzeugt, bzw. vernichtet. Die Messgröße ist wie in der klassischen Physik die Intensität, also das Betragsquadrat der elektrischen Feldstärke, bzw. der Erwartungswert davon.

Zur Orientierung zunächst eine Prinzipskizze:



Zum Vergleich die Skizze aus dem Original:



In den Quellen A und B werden Photonen erzeugt und in x_1 bzw. x_2 nachgewiesen ("vernichtet"). Verwendet man nur einen Detektor (z.B. in x_1), so spricht man von "Amplitudeninterferenz" oder mit Mandel von "Interferenz 2. Ordnung" (2. Ordnung, weil das Quadrat der E-Feldstärke gemessen wird) - also von der "ganz normalen Interferenz".

Misst man mit zwei Detektoren (in x_1 und x_2) in Koinzidenz, so spricht man von "Intensitätskorrelation" oder mit Mandel von "Interferenz 4. Ordnung" (die Messgröße ist das *Produkt* der Intensitäten). In jedem Fall bestimmen die Phasen F (bzw. die optischen Weglängen) das Interferenzmuster. Neben der von Mandel verwendeten Ordnung ist auch Interferenz 1. Ordnung (Amplitude) und Interferenz 2. Ordnung (Intensität) gebräuchlich. Wichtiger: Ob Interferenz beobachtbar ist, hängt von der Art der Quellen ab: A und B können primäre Quellen sein (z.B. Atome oder Laserstrahlen), oder sekundäre Quellen (z.B. Doppelspalt oder Strahlteiler), die von einer primären Quelle Q beleuchtet werden.

Im ersten Fall ist $|\Psi_{AB}\rangle$ ein Produktzustand, im zweiten Fall wird aus $|\Psi\rangle$ ein Superpositionszustand - mehr oder weniger kohärent bzw. verschränkt, und im Folgenden meist als "**Doppelspalt**" apostrophiert. Darüber hinaus spielt die Art des verwendeten Lichts eine wichtige Rolle. Die zwei wichtigsten Lichtsorten sind
a) Nummerzustände (**Fockzustände**) und
b) kohärente Zustände (**Glauberzustände**).
 Mit "Interferenz von Photonen" ist immer die Interferenz von Photonen in der gleichen Mode (Wellenlänge, Polarisation,...) gemeint.

Wir stellen zunächst die benötigten Operatoren und Zustände zusammen:

a) Fock

Vernichtungsoperatoren

> **ama:=Annihilation(A,1,explicit)**

$$ama := a_{A_1}^- \quad (1)$$

> **amb:=Annihilation(B,1,explicit)**

$$amb := a_{B_1}^- \quad (2)$$

und Erzeugungsoperatoren

> **cma:=ama^*; cmb:=amb^*;**

$$\begin{aligned} cma &:= a_{A_1}^+ \\ cmb &:= a_{B_1}^+ \end{aligned} \quad (3)$$

Zustände

> **KA:=Ket(A,n); KB:=Ket(B,m);**

$$\begin{aligned} KA &:= |A_n\rangle \\ KB &:= |B_m\rangle \end{aligned} \quad (4)$$

Abkürzung für den Zustand nach A und B

> **KAB:=Ket(Psi,AB)**

$$KAB := |\Psi_{AB}\rangle \quad (5)$$

Also z.B.

> **KAB=KA*KB**

$$|\Psi_{AB}\rangle = |A_n\rangle |B_m\rangle \quad (6)$$

Wirkung der Operatoren testen:

> **cma.ama.amb.KA.KB**

$$\sqrt{m} n |A_n\rangle |B_{m-1}\rangle \quad (7)$$

oder z.B.

> **(%Bracket=Bracket)(KA^*,cma.ama,KA*KB)**

$$\langle A_n | a_{A_1}^+ a_{A_1}^- | A_n \rangle | B_m \rangle = n | B_m \rangle \quad (8)$$

(Die linke Seite wird z.Zt. (Maple 2018) mit einem Ket zu viel geschrieben.)

Doppelspalt

Transformation des Zustands am Doppelspalt, der mit n Photonen beleuchtet wird:

> **DS:=n->1/sqrt(n!)*(1/sqrt(2))^n*(cma+cmb)^n.(Ket(A,0)*Ket(B,0))**

$$DS := n \mapsto \left(\frac{1}{\sqrt{n!}} \left(\frac{1}{\sqrt{2}} \right)^n (cma + cmb)^n \right) \cdot (|A_0\rangle |B_0\rangle) \quad (9)$$

> **DS(n)**

$$\frac{\left(\frac{\sqrt{2}}{2} \right)^n \left((a_{A_1}^+ + a_{B_1}^+)^n \cdot (|A_0\rangle |B_0\rangle) \right)}{\sqrt{n!}} \quad (10)$$

D.h., "Photonen werden am Doppelspalt binomial verteilt", z.B.:

> **KAB=DS(5)**

$$|\Psi_{AB}\rangle = \frac{\sqrt{360} |A_1\rangle |B_4\rangle}{48} + \frac{\sqrt{180} |A_2\rangle |B_3\rangle}{24} + \frac{\sqrt{180} |A_3\rangle |B_2\rangle}{24} + \frac{\sqrt{360} |A_4\rangle |B_1\rangle}{48} + \frac{\sqrt{2} (|A_0\rangle |B_5\rangle + |A_5\rangle |B_0\rangle)}{8} \quad (11)$$

Auf Anregung des Autors dieses Artikels wurde Anfang 2018 im Physics-Package die Option "hideketlabel" als Workaround eingeführt, mit der sich die übliche Kurzschreibweise realisieren lässt. Dann lautet der Zustand für

Ein Photon

> **KAB=DS(1)**

$$|AB\rangle = \frac{\sqrt{2} (|1\rangle |0\rangle + |0\rangle |1\rangle)}{2} \quad (12)$$

Zwei Photonen

> **KAB=DS(2)**

$$|AB\rangle = \frac{|0\rangle |2\rangle}{2} + \frac{|2\rangle |0\rangle}{2} + \frac{|1\rangle |1\rangle \sqrt{2}}{2} \quad (13)$$

(Anm.: $|1\rangle |1\rangle$ beschreibt ein "Biphoton" (zwei Photonen in zwei Moden), im Ggs. zum Zustand $|2\rangle$, also zwei Photonen in einer Mode.)

Fünf Photonen

> **KAB=DS(5)**

$$|AB\rangle = \frac{\sqrt{360} |1\rangle |4\rangle}{48} + \frac{\sqrt{180} |2\rangle |3\rangle}{24} + \frac{\sqrt{180} |3\rangle |2\rangle}{24} + \frac{\sqrt{360} |4\rangle |1\rangle}{48} + \frac{\sqrt{2} (|0\rangle |5\rangle + |5\rangle |0\rangle)}{8} \quad (14)$$

b) Glauber

> **KAc:=Ket(ama,alpha); KBc:=Ket(amb,beta);**

$$\begin{aligned} KAc &:= |a_{A_{1\alpha}}^- \rangle \\ KBc &:= |a_{B_{1\beta}}^- \rangle \end{aligned} \quad (15)$$

Test:

> **KAc^* .ama^* ,KBc^* .amb^***

$$\bar{\alpha} \langle a_{A_{1\alpha}}^- |, \bar{\beta} \langle a_{B_{1\beta}}^- | \quad (16)$$

Nicht gerade die übersichtlichste Notation...

Doppelspalt

> KA2c := Ket(ama, alpha/sqrt(2))

$$KA2c := \left| a_{A_1}^{-} \frac{\sqrt{2} \alpha}{2} \right\rangle \quad (17)$$

> ama.KA2c

$$\frac{\sqrt{2} \alpha}{2} \left| a_{A_1}^{-} \frac{\sqrt{2} \alpha}{2} \right\rangle \quad (18)$$

oder mit "versteckten Labeln":

> ama.KA2c

$$\frac{\sqrt{2} \alpha}{2} \left| \frac{\sqrt{2} \alpha}{2} \right\rangle \quad (19)$$

Wie bei klassischem Licht wird die Intensität halbiert (bei gleichen Amplituden am Doppelspalt).

Mit diesen Operatoren und Zuständen können wir folgende Erwartungswerte berechnen (Kopien aus dem Original):

the optical field at the two detectors. Then $P_1(x_1)$ and $P_2(x_1, x_2)$ are given by the expectations in normal order:

$$P_1(x_1) = \alpha_1 \langle \hat{E}^{(-)}(x_1) \hat{E}^{(+)}(x_1) \rangle, \quad (1)$$

$$P_2(x_1, x_2) = \alpha_1 \alpha_2 \langle \hat{E}^{(-)}(x_1) \hat{E}^{(-)}(x_2) \hat{E}^{(+)}(x_2) \hat{E}^{(+)}(x_1) \rangle, \quad (2)$$

where α_1, α_2 are constants characteristic of the detectors and the measurement times.

II. SECOND-ORDER INTERFERENCE

Let us decompose $\hat{E}^{(+)}(x_1)$ and $\hat{E}^{(+)}(x_2)$ into two normal modes A and B , such that \hat{a}_A, \hat{a}_B are the annihilation operators for the fields produced by the two sources S_A and S_B , respectively. Then we may put $\hat{E}^{(+)}(x_1) = f_A e^{i\phi_{A1}} \hat{a}_A + f_B e^{i\phi_{B1}} \hat{a}_B$, where f_A, f_B are complex parameters, and similarly for $\hat{E}^{(+)}(x_2)$. From Eq. (1) we then find

$$P_1(x_1) = \alpha_1 [|f_A|^2 \langle \hat{n}_A \rangle + |f_B|^2 \langle \hat{n}_B \rangle + f_A^* f_B e^{i(\phi_{B1} - \phi_{A1})} \langle \hat{a}_A^\dagger \hat{a}_B \rangle + \text{c.c.}] \quad (3)$$

Im Vergleich zu Mandels Artikel lassen wir im Folgenden die Ansprechwahrscheinlichkeiten α_1 und α_2 der Detektoren weg, bzw. setzen sie gleich 1.

Der Operator $E^{(+)}$ für das elektrische Feld (positiver Frequenzanteil) lautet dann:

> $E := i \rightarrow f_A[i] * a_{A1} + f_B[i] * a_{B1}$

$$E := i \mapsto f_{A_j} a_{A_j} + f_{B_j} a_{B_j} \quad (20)$$

also z.B.

> $'E(x[1])' = E(1)$

$$E(x_1) = f_{A_1} a_{A_1}^- + f_{B_1} a_{B_1}^- \quad (21)$$

mit den komplexen Amplituden f_{A_1} und f_{B_1} (sinngemäß für x_2).

Nach diesen Vorbereitungen nun zur Interferenz zweiter Ordnung:

Interferenz 2. Ordnung

a) Fock

Der Operator für das Quadrat der Feldstärke lautet

> $P1 := E(1)^* \cdot E(1)$

$$P1 := a_{A_1}^+ a_{A_1}^- |f_{A_1}|^2 + \overline{f_{A_1}} f_{B_1} a_{A_1}^+ a_{B_1}^- + a_{B_1}^+ a_{B_1}^- |f_{B_1}|^2 + \overline{f_{B_1}} f_{A_1} a_{B_1}^+ a_{A_1}^- \quad (22)$$

1. Unabhängige Quellen

Der Produktzustand mit n Photonen in der Mode A und m Photonen in der Mode B

> $KAB = K_A \cdot K_B$

$$|\Psi_{AB}\rangle = |A_n\rangle |B_m\rangle \quad (23)$$

hat dann den Erwartungswert

> (%Bracket=Bracket)(Dagger(KA.KB),P1,KA.KB)

$$\langle \langle B_m | \langle A_n | | a_{A_1}^+ a_{A_1}^- |f_{a_1}|^2 + \overline{f_{a_1}} f_{b_1} a_{A_1}^+ a_{B_1}^- + a_{B_1}^+ a_{B_1}^- |f_{b_1}|^2 + \overline{f_{b_1}} f_{a_1} a_{B_1}^+ a_{A_1}^- | | A_n \rangle | B_m \rangle \rangle = |f_{a_1}|^2 n + |f_{b_1}|^2 m \quad (24)$$

> EW1:=Dagger(KA.KB).P1.KA.KB

$$EW1 := |f_{a_1}|^2 n + |f_{b_1}|^2 m \quad (25)$$

D.h., man sieht keine Interferenz. Photonen in Nummerzuständen aus verschiedenen unabhängigen Quellen interferieren nicht, benehmen sich also wie klassisches inkohärentes Licht. Oder wie schon P.A.M. Dirac sagte: "Each photon then interferes only with itself. Interference between two different photons never occurs." (The Principles of Quantum Mechanics, 1957, S. 9)

2. Doppelspalt

Ein Photon aus Q (gleicher Abstand zu A und B) wird zu:

> KAB=DS(1)

$$|\Psi_{AB}\rangle = \frac{\sqrt{2}}{2} (|A_1\rangle |B_0\rangle + |A_0\rangle |B_1\rangle) \quad (26)$$

mit dem Erwartungswert in x1:

> DS1:=Dagger(DS(1)).P1.DS(1)

$$DS1 := \frac{\overline{f_{a_1}} f_{b_1}}{2} + \frac{|f_{a_1}|^2}{2} + \frac{|f_{b_1}|^2}{2} + \frac{\overline{f_{b_1}} f_{a_1}}{2} \quad (27)$$

Nun sieht man Interferenzterme (die gemischten Terme)

Wir schreiben die komplexen Amplituden in polarer Notation (Betrag und Phase):


> fas:=seq(fa[i]=f[A]*exp(I*phi[A]|i]),i=1..2); fbs:=seq(fb[i]=f[B]*exp(I*phi[B]|i]),i=1..2)

$$\begin{aligned} fas &:= fa_1 = f_A e^{i\phi_{A1}}, fa_2 = f_A e^{i\phi_{A2}} \\ fbs &:= fb_1 = f_B e^{i\phi_{B1}}, fb_2 = f_B e^{i\phi_{B2}} \end{aligned} \quad (28)$$

Und erhalten für die Intensität

> **IDS1:=combine(evalc(subs(fas,fbs,DS1)))**

$$IDS1 := f_A f_B \cos(\phi_{A1} - \phi_{B1}) + \frac{f_A^2}{2} + \frac{f_B^2}{2} \quad (29)$$




Also das bekannte Interferenzmuster für gleiche Amplituden (= 1 im Bild).

Die Sichtbarkeit als Animation:

> **eval(IDS1,[phi[B1]=0,f[A]=1])**

$$f_B \cos(\phi_{A1}) + \frac{1}{2} + \frac{f_B^2}{2} \quad (30)$$



Für 5 Photonen aus Q erhalten wir

> **DS5:=Dagger(DS(5)).P1.DS(5)**

$$DS5 := \frac{5 |fa_1|^2}{2} + \frac{5 |fb_1|^2}{2} + \frac{5 \overline{fa_1} fb_1}{2} + \frac{5 \overline{fb_1} fa_1}{2} \quad (31)$$

oder

> **combine(evalc(subs(fas, fbs, DS5)))**

$$\frac{5 f_A^2}{2} + \frac{5 f_B^2}{2} + 5 f_A f_B \cos(\phi_{A1} - \phi_{B1}) \quad (32)$$

Und es dürfte nicht schwer zu erraten sein, dass wir auch mit n Photonen das klassische Ergebnis erhalten. Warum? "Weil am Doppelspalt jedes Photon mit sich selbst interferiert!" (frei nach Dirac, s.o.). Das liegt daran, dass sich jedes Photon hinter dem Doppelspalt in einem Überlagerungszustand (oder sogar verschränkten Zustand - mit anderen Photonen und sich selbst) befindet.

b) Glauber

1. Unabhängige Quellen, z.B. zwei Laser "in der gleichen Mode":

Die Rechnung für kohärente Zustände lässt sich abkürzen, indem man die Operatoren durch ihre Eigenwerte ersetzt (im aktuellen Zustand (Juli 2018) des Physics-Package kann sie nicht mit dem Physics-Package durchgeführt werden):

> **EW1c:=(subs(ama=alpha,amb=beta,ama^*=conjugate(alpha),amb^*=conjugate(beta),P1))**

$$EW1c := \bar{\alpha} \alpha |fa_1|^2 + \overline{fa_1} fb_1 \bar{\alpha} \beta + \bar{\beta} \beta |fb_1|^2 + \overline{fb_1} fa_1 \bar{\beta} \alpha \quad (33)$$

> EW1c:=simplify(%)

$$EW1c := |fa_1|^2 |\alpha|^2 + \overline{fa_1} fb_1 \bar{\alpha} \beta + |fb_1|^2 |\beta|^2 + \overline{fb_1} fa_1 \bar{\beta} \alpha \quad (34)$$

Übergang zur polaren Notation

> EW1c:=combine(evalc(subs(fas,fbs,EW1c)))

$$EW1c := |\alpha|^2 f_A^2 - I \bar{\alpha} \sin(\phi_{A1} - \phi_{B1}) \beta f_A f_B + \bar{\alpha} \cos(\phi_{A1} - \phi_{B1}) \beta f_A f_B + |\beta|^2 f_B^2 + I \bar{\beta} \sin(\phi_{A1} - \phi_{B1}) \alpha f_A f_B + \bar{\beta} \cos(\phi_{A1} - \phi_{B1}) \alpha f_A f_B \quad (36)$$

(Ja, das ist z.Zt. alles noch etwas umständlich... :-)

Eigenwerte mit Betrag und Phase

> subs(alpha=abs(alpha)*exp(I*pa),beta=abs(beta)*exp(I*pb),EW1c)

$$\begin{aligned} & (|\alpha| e^{Ipa})^2 f_A^2 - I (\overline{|\alpha| e^{Ipa}}) \sin(\phi_{A1} - \phi_{B1}) |\beta| e^{Ipb} f_A f_B + (\overline{|\alpha| e^{Ipa}}) \cos(\phi_{A1} - \phi_{B1}) |\beta| e^{Ipb} f_A f_B + (|\beta| e^{Ipb})^2 f_B^2 + I (\overline{|\beta| e^{Ipb}}) \sin(\phi_{A1} \\ & - \phi_{B1}) |\alpha| e^{Ipa} f_A f_B + (\overline{|\beta| e^{Ipb}}) \cos(\phi_{A1} - \phi_{B1}) |\alpha| e^{Ipa} f_A f_B \end{aligned} \quad (37)$$

> evalc(%)

$$\begin{aligned} & (|\alpha|^2 f_A^2 + |\beta|^2 f_B^2 + 2 (-|\alpha| \sin(pa) \sin(\phi_{A1} - \phi_{B1}) |\beta| \cos(pb) + |\alpha| \cos(pa) \sin(\phi_{A1} - \phi_{B1}) |\beta| \sin(pb)) f_A f_B + 2 (|\alpha| \cos(pa) \cos(\phi_{A1} \\ & - \phi_{B1}) |\beta| \cos(pb) + |\alpha| \sin(pa) \cos(\phi_{A1} - \phi_{B1}) |\beta| \sin(pb)) f_A f_B + I ((-|\alpha| \cos(pa) \sin(\phi_{A1} - \phi_{B1}) |\beta| \cos(pb) \\ & - |\alpha| \sin(pa) \sin(\phi_{A1} - \phi_{B1}) |\beta| \sin(pb)) f_A f_B + (-|\alpha| \sin(pa) \cos(\phi_{A1} - \phi_{B1}) |\beta| \cos(pb) + |\alpha| \cos(pa) \cos(\phi_{A1} \\ & - \phi_{B1}) |\beta| \sin(pb)) f_A f_B + (|\alpha| \cos(pa) \sin(\phi_{A1} - \phi_{B1}) |\beta| \cos(pb) + |\alpha| \sin(pa) \sin(\phi_{A1} - \phi_{B1}) |\beta| \sin(pb)) f_A f_B + (\\ & -|\alpha| \cos(pa) \cos(\phi_{A1} - \phi_{B1}) |\beta| \sin(pb) + |\alpha| \sin(pa) \cos(\phi_{A1} - \phi_{B1}) |\beta| \cos(pb)) f_A f_B \end{aligned} \quad (38)$$

> Intensc:=combine(%)

$$Intensc := |\alpha|^2 f_A^2 + |\beta|^2 f_B^2 + 2 |\alpha| |\beta| f_A f_B \cos(-pb + pa + \phi_{A1} - \phi_{B1}) \quad (39)$$

Mit dem Phasenunterschied δ der Zustände ($|\alpha|^2$ und $|\beta|^2$ sind die mittleren Photonenzahlen) erhält man:

> Intenscd := abs(alpha)^2*(f[A]^2)+abs(beta)^2*(f[B]^2)+2*abs(alpha*beta)*f[A]*f[B]*cos(phi[A1]-phi[B1]+delta)

$$Intenscd := |\alpha|^2 f_A^2 + |\beta|^2 f_B^2 + 2 |\alpha| |\beta| f_A f_B \cos(\phi_{A1} - \phi_{B1} + \delta) \quad (40)$$

Na endlich:

Photonen aus verschiedenen Quellen in kohärenten Zuständen interferieren ganz klassisch! Allerdings handelt es sich hier um Momentanwerte, d.h., die Amplituden (α und β) und insbes. die Phasendifferenz δ sind Funktionen der Zeit und das Interferenzmuster ist nur gut sichtbar, wenn die Laser stabil arbeiten (wie bei klassischen Quellen auch).

Zum Vergleich noch einmal das "klassische Ergebnis":

$$f_A f_B \cos(\phi_{A1} - \phi_{B1}) + \frac{f_A^2}{2} + \frac{f_B^2}{2} \quad (42)$$

Also lag Dirac falsch mit seinem Statement "Interference between two different photons never occurs"? Nein, Dirac sagte:

"Some time before the discovery of quantum mechanics people realized that the connexion between light waves and photons must be of a statistical character. What they did not clearly realize, however, was that the wave function gives information about the probability of one photon being in a particular place and not the probable number of photons in that place. The importance of the distinction can be made clear in the following way. Suppose we have a beam of light consisting of a large number of photons split up into two components of equal intensity. On the assumption that the intensity of a beam is connected with the probable number of photons in it, we should have half the total number of photons going into each component. If the two components are now made to interfere, we should require a photon in one component to be able to interfere with one in the other. Sometimes these two photons would have to annihilate one another and other times they would have to produce four photons. This would contradict the conservation of energy. The new theory, which connects the wave function with the probabilities for one photon, gets over the difficulty by making each photon go partly into each of the two components. Each photon then interferes only with itself. Interference between two different photons never occurs."

2. Doppelspalt

Der Doppelspalt kann als kohärente Quelle im Produktzustand

> **KAB=KA2c*KA2c**

$$|\Psi_{AB}\rangle = \left| a_{A_1}^{-} \frac{\sqrt{2}\alpha}{2} \right\rangle \left| a_{A_1}^{-} \frac{\sqrt{2}\alpha}{2} \right\rangle \quad (43)$$

Kurznotation:

> **KAB=KA2c*KA2c**

$$|AB\rangle = \left| \frac{\sqrt{2}\alpha}{2} \right\rangle \left| \frac{\sqrt{2}\alpha}{2} \right\rangle \quad (44)$$

behandelt werden. Man erhält für die Intensität

> **subs(alpha=alpha/sqrt(2),beta=alpha/sqrt(2),pa=0,pb=0,Intensc)**

$$\left| \frac{\sqrt{2}\alpha}{2} \right|^2 f_A^2 + \left| \frac{\sqrt{2}\alpha}{2} \right|^2 f_B^2 + 2 \left| \frac{\alpha^2}{2} \right| f_A f_B \cos(\phi_{A1} - \phi_{B1}) \quad (45)$$

oder

> **simplify(%)**

$$\frac{|\alpha|^2 \left(2 f_A f_B \cos(\phi_{A1} - \phi_{B1}) + f_A^2 + f_B^2 \right)}{2} \quad (46)$$

Richtig: Beleuchtet man einen Doppelspalt mit Licht im kohärenten Zustand $|\alpha\rangle$, so sieht man "klassische Interferenz" :-).

Zusammenfassung

a) Fock

1. Unabhängige (primäre) Quellen: Photonen in Fockzuständen interferieren nicht.
2. Sekundäre Quellen (Doppelspalt, Strahlteiler,...): Photonen aus Fockzuständen (aus einer Quelle) können interferieren (mit sich selbst).

b) Glauber

1. Unabhängige (primäre) Quellen: Photonen in Glauberzuständen können interferieren (auch über längere Zeit, wenn man die Quellen stabil hält).
2. Doppelspalt: Photonen aus Glauberzuständen interferieren - quasi klassisch.

Die zugehörigen Gleichungen lauten:

Operator Zustand	Operator mikomma.de
	$P(x_1) = f_A ^2 \langle \hat{n}_A \rangle + f_B ^2 \langle \hat{n}_B \rangle + 2f_A^* f_B e^{i(\phi_B - \phi_A)} \langle \hat{a}_A^\dagger \hat{a}_B \rangle + c.c.$
$ n, m\rangle$	keine Interferenz: $n f_A ^2 + m f_B ^2$
Doppelspalt mit einem Photon: $ 1, 0\rangle + 0, 1\rangle$	$\frac{1}{2} (f_A^2 + f_B^2 + 2f_A f_B \cos(\phi_A - \phi_B))$, vgl. 4.16 auf Seite 162 $\langle 1, 0 \hat{a}_A^\dagger \hat{a}_A 1, 0 \rangle + \langle 0, 1 \hat{a}_B^\dagger \hat{a}_B 0, 1 \rangle + 2 \langle 1, 0 \hat{a}_A^\dagger \hat{a}_B 0, 1 \rangle \cos(\phi_A - \phi_B)$
n Photonen $(\hat{a}_A^\dagger + \hat{a}_B^\dagger)^n 0, 0\rangle$	$\frac{n}{2} (f_A^2 + f_B^2 + 2f_A f_B \cos(\phi_A - \phi_B))$
$ \alpha, \beta\rangle$	$ \alpha ^2 f_A^2 + \beta ^2 f_B^2 + 2 \alpha\beta f_A f_B \cos(\phi_A - \phi_B + \delta)$
$ \frac{\alpha}{\sqrt{2}}, \frac{\alpha}{\sqrt{2}}\rangle$	$\frac{ \alpha ^2}{2} (f_A^2 + f_B^2 + 2f_A f_B \cos(\phi_A - \phi_B))$

Interferenz 4. Ordnung

Zur Orientierung der Erwartungswert aus Mandels Artikel:

S276 L. Mandel: Quantum effects in photon interference

$$\begin{aligned}
 P_2(x_1, x_2) = & \alpha_1 \alpha_2 \{ |f_A|^4 \langle \hat{n}_A^2 \rangle + |f_B|^4 \langle \hat{n}_B^2 \rangle + 2|f_A|^2 |f_B|^2 \langle \hat{n}_A \rangle \langle \hat{n}_B \rangle [1 + \cos(\phi_{B2} - \phi_{A2} + \phi_{A1} - \phi_{B1})] \\
 & + f_A^{*2} f_B^2 \langle \hat{a}_A^{\dagger 2} \hat{a}_B^2 \rangle e^{i(\phi_{B2} - \phi_{A2} + \phi_{B1} - \phi_{A1})} + c.c. + |f_A|^2 f_A^* f_B \langle \hat{a}_A^{\dagger 2} \hat{a}_A \hat{a}_B \rangle [e^{i(\phi_{B1} - \phi_{A1})} + e^{i(\phi_{B2} - \phi_{A2})}] + c.c. \\
 & + |f_B|^2 f_B^* f_A \langle \hat{a}_B^{\dagger 2} \hat{a}_B \hat{a}_A \rangle [e^{i(\phi_{A1} - \phi_{B1})} + e^{i(\phi_{A2} - \phi_{B2})}] + c.c. \}
 \end{aligned}
 \tag{5}$$

Der Operator lautet mit unseren Definitionen von oben:

> **P2 := subsindets(E(1)^* . E(2)^* . E(2) . E(1), specfunc(conjugate), expand)**

$$\begin{aligned}
 P2 := & \overline{fb_2} \overline{fa_2} a^+_{A_1} a^+_{B_1} (a^-_{A_1})^2 |fa_1|^2 + |fb_2|^2 a^+_{A_1} a^+_{B_1} a^-_{B_1} a^-_{A_1} |fa_1|^2 + \overline{fa_2} \overline{fb_2} (a^+_{A_1})^2 a^-_{B_1} a^-_{A_1} |fa_1|^2 + |fa_2|^2 (a^+_{A_1})^2 (a^-_{A_1})^2 |fa_1|^2 \\
 & + \overline{fa_1} \overline{fb_1} |fa_2|^2 (a^+_{A_1})^2 a^-_{A_1} a^-_{B_1} + \overline{fa_1} \overline{fa_2} \overline{fb_1} \overline{fb_2} (a^+_{A_1})^2 (a^-_{B_1})^2 + \overline{fa_1} \overline{fb_1} |fb_2|^2 a^+_{A_1} a^+_{B_1} (a^-_{B_1})^2 + \overline{fa_1} \\
 & \overline{fb_2} \overline{fa_2} \overline{fb_1} a^+_{A_1} a^+_{B_1} a^-_{A_1} a^-_{B_1} + \overline{fb_2} \overline{fa_2} (a^+_{B_1})^2 a^-_{A_1} a^-_{B_1} |fb_1|^2 + |fb_2|^2 (a^+_{B_1})^2 (a^-_{B_1})^2 |fb_1|^2 + |fa_2|^2 a^+_{B_1} a^+_{A_1} a^-_{A_1} a^-_{B_1} |fb_1|^2 + \\
 & \overline{fa_2} \overline{fb_2} a^+_{B_1} a^+_{A_1} (a^-_{B_1})^2 |fb_1|^2 + \overline{fb_1} \overline{fa_1} |fa_2|^2 a^+_{B_1} a^+_{A_1} (a^-_{A_1})^2 + \overline{fb_1} \overline{fa_2} \overline{fa_1} \overline{fb_2} a^+_{B_1} a^+_{A_1} a^-_{B_1} a^-_{A_1} + \\
 & \overline{fb_1} \overline{fa_1} |fb_2|^2 (a^+_{B_1})^2 a^-_{B_1} a^-_{A_1} + \overline{fb_1} \overline{fb_2} \overline{fa_1} \overline{fa_2} (a^+_{B_1})^2 (a^-_{A_1})^2
 \end{aligned} \quad (47)$$

a) Fock

1. n Photonen in der Mode A, m Photonen in der Mode B

Erwartungswert der Koinzidenzzählrate in χ_1 und χ_2 für den Produktzustand

> **KAB=KA*KB**

$$|\Psi_{AB}\rangle = |A_n\rangle |B_m\rangle \quad (48)$$

> **EW2:=Dagger(KA.KB).P2.KA.KB**

$$EW2 := m n \overline{fa_2} \overline{fb_1} (\overline{fa_1} \overline{fb_2}) + m n \overline{fa_1} \overline{fb_2} (\overline{fa_2} \overline{fb_1}) + ((n-1) |fa_2|^2 + |fb_2|^2 m) n |fa_1|^2 + (|fa_2|^2 n + |fb_2|^2 (m-1)) |fb_1|^2 m \quad (49)$$

Mit der gleichen Prozedur wie oben (Interferenz 2. Ordnung) erhalten wir (gekürzte Fassung):

Amplituden f mit Betrag und Phase

> **EW2 := 2*m*n*(f[A]^2)*(f[B]^2)*cos(-phi[A2]-phi[B1]+phi[A1]+phi[B2])-n^2*(f[A]^4)-m^2*(f[B]^4)-m*(f[B]^4)**

$$EW2 := 2 m n f_A^2 f_B^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + 2 m n f_A^2 f_B^2 + n^2 f_A^4 - n f_A^4 + m^2 f_B^4 - m f_B^4 \quad (50)$$

Die Photonenzahlen m und n entsprechen den Amplitudenquadraten f^2 . Wir untersuchen die Sichtbarkeit der Interferenz 4. Ordnung am Beispiel des Zustands $|1\rangle|1\rangle$

2. Jeweils ein Photon in A und B

Mandels Gleichung (5) ergibt mit dem Zustand

> **Ket(Psi, AB) = Ket(A, 1)*Ket(B, 1)**

$$|\Psi_{AB}\rangle = |A_1\rangle |B_1\rangle \quad (51)$$

die Koinzidenzzählrate


> **Intens2:=subs(n=1,m=1,EW2)**

$$Intens2 := 2 f_A^2 f_B^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + 2 f_A^2 f_B^2 \quad (52)$$

Und das ist nicht die klassische Interferenzformel (egal welcher Ordnung)! Hier ein animiertes Beispiel zur Sichtbarkeit:

> $I_{2ani} := \text{subs}(\text{phi}[A2]=0, \text{phi}[B2]=0, \text{phi}[B1]=0, f[A]=1, I_{\text{Intens2}})$

$$I_{2ani} := 2 f_B^2 \cos(\phi_{A1}) + 2 f_B^2 \quad (53)$$



Das ist ein "echter Quanteneffekt": Die Minima der Interferenz 4. Ordnung liegen immer bei 0, bzw. die Sichtbarkeit der Interferenz 4. Ordnung ist immer 100% (im Gegensatz zur Sichtbarkeit der Interferenz 2. Ordnung), vor allem aber im Gegensatz zur Intensitätskorrelation von klassischem Licht (s.u.).

3. Doppelspalt

Superpositionszustände von Fockzuständen unterscheiden sich bei Koinzidenzmessungen nicht wesentlich von Glauberzuständen. Deshalb hier nur ein Beispiel für den Fall, dass "ein Doppelspalt mit 5 Photonen beleuchtet wird". Produkt der Intensitäten:

$$10 f_A^2 f_B^2 \cos(\phi_{A1} + \phi_{B2} - \phi_{A2} - \phi_{B1}) + 10 f_A^2 f_B^2 \cos(-\phi_{B1} - \phi_{B2} + \phi_{A1} + \phi_{A2}) + 5 (f_A^2 + f_B^2) (2 f_A f_B \cos(\phi_{A1} - \phi_{B1}) + 2 f_A f_B \cos(-\phi_{B2} + \phi_{A2}) + f_A^2 + f_B^2)$$

Zum Vergleich mit Gleichung (57) unten. Bei der Beleuchtung mit einem Photon ist die Koinzidenzzählrate natürlich Null :).

b) Glauber

1. Momentanwerte

Wir berechnen den Erwartungswert der Koinzidenzzählrate für den Zustand

> $K_{AB} = K_{Ac} * K_{Bc}$

$$|AB\rangle = |\alpha\rangle |\beta\rangle \quad (54)$$

Der Operator P2 von oben lautet mit den Eigenwerten α und β :

> **EW2c:=subs(ama=alpha,amb=beta,ama^*=conjugate(alpha),amb^*=conjugate(beta),P2)**

$$\begin{aligned}
 EW2c := & \overline{fb_2} \overline{fa_2} \overline{\alpha} \overline{\beta} \alpha^2 |fa_1|^2 + |fb_2|^2 \overline{\alpha} \overline{\beta} \beta \alpha |fa_1|^2 + \overline{fa_2} \overline{fb_2} \overline{\alpha}^2 \beta \alpha |fa_1|^2 + |fa_2|^2 \overline{\alpha}^2 \alpha^2 |fa_1|^2 + \overline{fa_1} \overline{fb_1} |fa_2|^2 \overline{\alpha}^2 \alpha \beta + \overline{fa_1} \overline{fa_2} \overline{fb_1} \overline{fb_2} \overline{\alpha}^2 \beta^2 + \\
 & \overline{fa_1} \overline{fb_1} |fb_2|^2 \overline{\alpha} \overline{\beta} \beta^2 + \overline{fa_1} \overline{fb_2} \overline{fa_2} \overline{fb_1} \overline{\alpha} \overline{\beta} \alpha \beta + \overline{fb_2} \overline{fa_2} \overline{\beta}^2 \alpha \beta |fb_1|^2 + |fb_2|^2 \overline{\beta}^2 \beta^2 |fb_1|^2 + |fa_2|^2 \overline{\beta} \overline{\alpha} \alpha \beta |fb_1|^2 + \overline{fa_2} \overline{fb_2} \overline{\beta} \overline{\alpha} \beta^2 |fb_1|^2 + \\
 & \overline{fb_1} \overline{fa_1} |fa_2|^2 \overline{\beta} \overline{\alpha} \alpha^2 + \overline{fb_1} \overline{fa_2} \overline{fa_1} \overline{fb_2} \overline{\beta} \overline{\alpha} \beta \alpha + \overline{fb_1} \overline{fa_1} |fb_2|^2 \overline{\beta}^2 \beta \alpha + \overline{fb_1} \overline{fb_2} \overline{fa_1} \overline{fa_2} \overline{\beta}^2 \alpha^2
 \end{aligned} \quad (55)$$

Mit der Phasenverschiebung d der kohärenten Zustände ergibt sich für den Erwartungswert nach einer Reihe von Umformungen (zu denen man Maple überreden muss):

$$\begin{aligned}
 X := & 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(-2\delta + \phi_{A1} + \phi_{A2} - \phi_{B1} - \phi_{B2}) + 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + 2 |\beta| |\alpha|^3 \cos(-\delta + \phi_{A2} - \phi_{B2}) f_A^3 f_B \\
 & + 2 |\beta| |\alpha|^3 \cos(-\delta + \phi_{A1} - \phi_{B1}) f_A^3 f_B + 2 |\beta|^3 |\alpha| \cos(-\delta + \phi_{A2} - \phi_{B2}) f_A f_B^3 + 2 |\beta|^3 |\alpha| \cos(-\delta + \phi_{A1} - \phi_{B1}) f_A f_B^3 + 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 + f_A^4 |\alpha|^4 \\
 & + f_B^4 |\beta|^4
 \end{aligned} \quad (56)$$

In diesem Term (X) bestimmen die Amplituden f , $|\alpha|$ und $|\beta|$ die Sichtbarkeit. Die Phasen F_{A2} und F_{B2} bestimmen die Amplitude und F_{B1} die Verschiebung und die Phase d die Amplitude und Verschiebung. Wir wollen hier nur kurz andeuten, wie sich eine Phasenverschiebung zweier "identischer Laser" auf das beobachtete Interferenzmuster (4. Ordnung) auswirkt:


> **I4c:=simplify(subs(alpha=1,beta=1,X))**

$$\begin{aligned}
 I4c := & 2 f_A^2 f_B^2 \cos(2\delta - \phi_{A1} - \phi_{A2} + \phi_{B1} + \phi_{B2}) + 2 f_A^2 f_B^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + (f_A^2 + f_B^2) (2 f_A f_B \cos(\delta - \phi_{A1} + \phi_{B1}) + 2 f_A f_B \cos(\delta \\
 & - \phi_{A2} + \phi_{B2}) + f_A^2 + f_B^2)
 \end{aligned} \quad (57)$$

Animation mit d :

> **I4cani:=subs(f[A]=1, f[B]=1, phi[B1]=0, phi[B2]=0, phi[A2]=0,X)**

$$I4cani := 2 \cos(2\delta - \phi_{A1}) + 2 \cos(\phi_{A1}) + 4 \cos(\delta - \phi_{A1}) + 4 \cos(\delta) + 4 \quad (58)$$



Man kann die Quellen also auch so einstellen, dass man nichts sieht :).

2. Mittelung für unkorrelierte Quellen in kohärenten Zuständen

Wenn sich die Phasenverschiebung d der kohärenten Zustände nicht kontrollieren lässt, muss man über sie mitteln. Dann erhält man für die Koinzidenzzählrate

> **I4cu=subs(pb=0,pa=0,EWs)**

$$\begin{aligned}
 I4cu = & 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(\phi_{A1} + 2\delta + \phi_{A2} - \phi_{B1} - \phi_{B2}) + 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + 2 |\beta| |\alpha|^3 \cos(\phi_{A2} + \delta - \phi_{B2}) f_A^3 f_B \\
 & + 2 |\beta| |\alpha|^3 \cos(\phi_{A1} + \delta - \phi_{B1}) f_A^3 f_B + 2 |\beta|^3 |\alpha| \cos(\phi_{A2} + \delta - \phi_{B2}) f_A f_B^3 + 2 |\beta|^3 |\alpha| \cos(\phi_{A1} + \delta - \phi_{B1}) f_A f_B^3 + 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 + f_A^4 |\alpha|^4 + \\
 & f_B^4 |\beta|^4 \quad (59)
 \end{aligned}$$

Mittelung (Integration über d):

> **int(EWs,delta=0..2*Pi)/(2*Pi)**

$$\begin{aligned}
 & \frac{1}{2\pi} \left(4 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \pi + 2 f_A^4 |\alpha|^4 \pi + 2 f_B^4 |\beta|^4 \pi + 4 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \pi \cos(\phi_{A1}) \cos(\phi_{A2}) \sin(\phi_{B1}) \sin(\phi_{B2}) + 4 f_A^2 \right. \\
 & f_B^2 |\alpha|^2 |\beta|^2 \pi \sin(\phi_{A1}) \cos(\phi_{A2}) \cos(\phi_{B2}) \sin(\phi_{B1}) - 4 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \pi \sin(\phi_{A1}) \cos(\phi_{A2}) \cos(\phi_{B1}) \sin(\phi_{B2}) + 4 f_A^2 \\
 & f_B^2 |\alpha|^2 |\beta|^2 \pi \sin(\phi_{A1}) \sin(\phi_{A2}) \cos(\phi_{B1}) \cos(\phi_{B2}) + 4 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \pi \sin(\phi_{A1}) \sin(\phi_{A2}) \sin(\phi_{B1}) \sin(\phi_{B2}) - 4 f_A^2 \\
 & f_B^2 |\alpha|^2 |\beta|^2 \pi \cos(\phi_{A1}) \sin(\phi_{A2}) \cos(\phi_{B2}) \sin(\phi_{B1}) + 4 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \pi \cos(\phi_{A1}) \sin(\phi_{A2}) \cos(\phi_{B1}) \sin(\phi_{B2}) + 4 f_A^2 \\
 & \left. f_B^2 |\alpha|^2 |\beta|^2 \pi \cos(\phi_{A1}) \cos(\phi_{A2}) \cos(\phi_{B1}) \cos(\phi_{B2}) \right) \quad (60)
 \end{aligned}$$

> **combine(%)**

$$2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + f_A^4 |\alpha|^4 + 2 f_A^2 f_B^2 |\alpha|^2 |\beta|^2 + f_B^4 |\beta|^4 \quad (61)$$

Was dem "klassischen Ergebnis" entspricht ($\alpha = \beta = 1$):

$$> I4klass := 2 * f[A]^2 * f[B]^2 * \cos(-\text{phi}[A2] - \text{phi}[B1] + \text{phi}[B2] + \text{phi}[A1]) + (f[A]^2 + f[B]^2)^2$$

$$I4klass := 2 f_A^2 f_B^2 \cos(-\phi_{A2} - \phi_{B1} + \phi_{A1} + \phi_{B2}) + (f_A^2 + f_B^2)^2 \quad (67)$$

Siehe auch die Box A im Original:

Box A: Comparison of quantum and classical fourth-order interference.

For the case in which exactly one photon is emitted by S_A and one photon by S_B , we have

$$\langle : \hat{n}_A^2 : \rangle = \langle \hat{n}_A (\hat{n}_A - 1) \rangle = 0, \quad \langle : \hat{n}_B^2 : \rangle = \langle \hat{n}_B (\hat{n}_B - 1) \rangle = 0,$$

$$\langle \hat{a}_A^{\dagger 2} \hat{a}_B^2 \rangle = 0, \quad \langle \hat{a}_A^{\dagger 2} \hat{a}_A \hat{a}_B \rangle = 0, \quad \langle \hat{a}_B^{\dagger 2} \hat{a}_B \hat{a}_A \rangle = 0.$$

Therefore of all the terms on the right hand side of Eq. (5) only the third survives, and we obtain finally,

$$P_2(x_1, x_2) = \alpha_1 \alpha_2 2 |f_A|^2 |f_B|^2 [1 + \cos(\phi_{B2} - \phi_{A2} + \phi_{A1} - \phi_{B1})]. \quad (6)$$

This result exhibits two-photon interference with 100% visibility.

Let us contrast this conclusion for a two-photon state with the result given by the same Eq. (5) for a classical state of the incoming field, when the two sources are completely independent. In this case we have to treat \hat{a}_A, \hat{a}_B as complex c -number amplitudes, $\langle \hat{n}_A \rangle$ becomes the mean light intensity $\langle I_A \rangle$, and $\langle : \hat{n}_A^2 : \rangle$ becomes $\langle I_A^2 \rangle$. Because of the phase independence, all terms on the right hand side of Eq. (5) beyond the first three vanish again, but this time terms one and two are nonzero, and we have the result


$$P_2(x_1, x_2) = \alpha_1 \alpha_2 2 [(|f_A|^2 \langle I_A \rangle + |f_B|^2 \langle I_B \rangle)^2 + 2 |f_A|^2 |f_B|^2 \langle I_A \rangle \langle I_B \rangle \cos(\phi_{B2} - \phi_{A2} + \phi_{A1} - \phi_{B1})].$$

It is not difficult to prove that in this classical field case the visibility of the interference has an upper bound of 1/2, compared with the value 1 given by Eq. (6) for the case of a quantum field (Richter, 1979).

Animation zur Sichtbarkeit

$$> I4klani := \text{subs}(f[A]=1, \text{phi}[A2]=0, \text{phi}[B1]=0, \text{phi}[B2]=0, I4klass)$$

$$I4klani := 2 f_B^2 \cos(\phi_{A1}) + (f_B^2 + 1)^2 \quad (68)$$



Im Gegensatz zu Fockzuständen (zwei Photonen) liegen die Minima nicht bei 0 und die maximale Sichtbarkeit beträgt nur 50%!

Zusammenfassung

a) Fock

1. Für die Zustände $|n\rangle|m\rangle$ (mit $n, m > 1$) ergeben sich "quasiklassische" Intensitätskorrelationen.
2. Für den Zustand $|1\rangle|1\rangle$ kann man einen "echten Quanteneffekt" beobachten: die Sichtbarkeit des Interferenzmusters wird 1!

b) Glauber

1. Mit kontrollierbaren Glauber-Produktzuständen kann man "beliebige" Intensitätskorrelationen erhalten :)
2. Wenn sich die Phase nicht kontrollieren lässt - wie bei "klassischem/thermischem Licht" - ist die maximale Sichtbarkeit des Interferenzmusters nur 1/2.

Die zugehörigen Gleichungen lauten:

$$\begin{aligned}
 P_2(x_1, x_2) = & |f_A|^4 \langle : \hat{n}_A^2 : \rangle + |f_B|^4 \langle : \hat{n}_B^2 : \rangle + \\
 & 2|f_A|^2 |f_B|^2 \langle \hat{n}_A \rangle \langle \hat{n}_B \rangle [1 + \cos(\phi_{B2} - \phi_{A2} + \phi_{A1} - \phi_{B1})] + \\
 & f_A^{*2} f_B^2 \langle \hat{a}_A^{\dagger 2} \hat{a}_B^2 \rangle e^{i(\phi_{B2} - \phi_{A2} + \phi_{B1} - \phi_{A1})} + c.c. + \\
 & |f_A|^2 f_A^* f_B \langle \hat{a}_A^{\dagger 2} \hat{a}_A \hat{a}_B \rangle \left[e^{i(\phi_{B1} - \phi_{A1})} + e^{i(\phi_{B2} - \phi_{A2})} \right] + c.c. + \\
 & |f_B|^2 f_B^* f_A \langle \hat{a}_B^{\dagger 2} \hat{a}_B \hat{a}_A \rangle \left[e^{i(\phi_{A1} - \phi_{B1})} + e^{i(\phi_{A2} - \phi_{B2})} \right] + c.c.
 \end{aligned}$$

n Photonen in A, m Photonen in B:

$$\begin{aligned}
 |n\rangle |m\rangle \rightarrow & n^2 f_A^4 - n f_A^4 + m^2 f_B^4 - m f_B^4 + 2nm f_A^2 f_B^2 \\
 & + 2nm f_A^2 f_B^2 \cos(\phi_{A1} - \phi_{A2} + \phi_{B2} - \phi_{B1})
 \end{aligned}$$

n=m=1:

$$|1\rangle |1\rangle \rightarrow 2f_A^2 f_B^2 + 2f_A^2 f_B^2 \cos(\phi_{A1} - \phi_{A2} + \phi_{B2} - \phi_{B1})$$

Glauber, $|\alpha\rangle |\beta\rangle$ mit Phasenunterschied δ :

$$\begin{aligned}
 |\alpha\rangle |\beta\rangle \rightarrow & f_A^4 |\alpha|^4 + f_B^4 |\beta|^4 + 2f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \\
 & + 2f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(\phi_{A1} - \phi_{A2} + \phi_{B2} - \phi_{B1}) \\
 & + 2f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(\phi_{A1} + \phi_{A2} - \phi_{B2} - \phi_{B1} - 2\delta) \\
 & + 2f_A^3 f_B |\alpha|^3 |\beta| \cos(\phi_{A1} - \phi_{B1} - \delta) \\
 & + 2f_A^3 f_B |\alpha|^3 |\beta| \cos(\phi_{A2} - \phi_{B2} - \delta) \\
 & + 2f_A f_B^3 |\alpha| |\beta|^3 \cos(\phi_{A1} - \phi_{B1} - \delta) \\
 & + 2f_A f_B^3 |\alpha| |\beta|^3 \cos(\phi_{A2} - \phi_{B2} - \delta)
 \end{aligned}$$

Bei Mittelung über δ entfallen die Terme mit δ : mikomma.de

$$f_A^4 |\alpha|^4 + f_B^4 |\beta|^4 + 2f_A^2 f_B^2 |\alpha|^2 |\beta|^2 + 2f_A^2 f_B^2 |\alpha|^2 |\beta|^2 \cos(\phi_{A1} - \phi_{A2} + \phi_{B2} - \phi_{B1})$$

© Juli 2018, Dr. Michael Komma (VGWORT)

Siehe auch:

[Matrixoptik](#) | [Doppelspalt und Dualismus](#) | [Quantenradierer](#) | [Photon am Doppelspalt](#) | [Gitter](#) | [Punktgitter](#) | [Kreuzgitter](#) | [Raumgitter](#) | [Beugung](#) | [Fresnelbeugung](#) | [Zeiger](#) | [Interferenz](#)

[Moderne Physik mit Maple](#)

komma_AT_oe.uni-tuebingen.de

[HOME](#) | [Fächer](#) | [Physik](#) | [Elektrizität](#) | [Optik](#) | [Atomphysik](#) | [Quantenphysik](#) | [Top](#)